

# *Scattering and Gradient Forces from the Electromagnetic Stress Tensor Acting on a Dielectric Sphere*

**Zachary H. Levine**  
**J. J. Curry**

The derivation of the scattering force and the gradient force on a spherical particle due to an electromagnetic wave often invokes the Clausius–Mossotti factor, based on an ad hoc physical model. In this article, we derive the expressions including the Clausius–Mossotti factor directly from the fundamental equations of classical electromagnetism. Starting from an analytic expression for the force on a spherical particle in a vacuum using the Maxwell stress tensor, as well as the Mie solution for the response of dielectric particles to an electromagnetic plane wave, we derive the scattering and gradient forces. In both cases, the Clausius–Mossotti factor arises rigorously from the derivation without any physical argumentation. The limits agree with expressions in the literature.

## ■ 1. Overview

Recently, we made a theoretical study of a system to sort submicrometer dielectric spheres in the interference field of a laser in slowly flowing air [1, 2]. In the course of that project, we derived the scattering and gradient forces rigorously from Maxwell’s equations. The derivation was too detailed for that article, so we are presenting it here. The results agree with expressions from Harada and Asakura [3], as we show in the Appendix. In addition, we present the code we used to derive the force on a spherical particle used in [1, 2].

In this article, we develop code to generate the Mie scattering coefficients and the stress tensor formulas, and we combine them to form first the scattering force and then the gradient force. The scattering force comes first because it requires an incident plane wave, whereas the gradient force requires an incident standing wave that is a little more difficult to set up.

The solution for the response of a spherical dielectric particle in a vacuum was given more than 100 years ago [4]. The problem has been studied extensively by Bohren and Huffman [5]. Our formulation follows the textbook of Zangwill [6]. Since the problem is so well studied, we go directly to the solution. We omit an implicit time dependence  $\exp(-i\omega t)$  in the plane wave traveling in the positive  $z$  direction with the electric field linearly polarized in the  $x$  direction. The electric field  $\mathbf{E}$  and magnetic field  $\mathbf{B}$  are thereby given by

$$\mathbf{E} = -k^{-1} \nabla \times (k \mathbf{r} u) + i k^{-2} \nabla \times [\nabla \times (k \mathbf{r} w)], \quad (1)$$

$$c \mathbf{B} = k^{-1} \nabla \times (k \mathbf{r} w) + i k^{-2} \nabla \times [\nabla \times (k \mathbf{r} u)], \quad (2)$$

where  $u$  and  $w$  are scalar functions related to the transverse electric and transverse magnetic parts of the solution, respectively,  $c$  is the speed of light,  $k = \omega c^{-1}$  and  $\mathbf{r}$  is a point in space. The International System of Units (SI) is used; in these units  $c \mathbf{B}$  has the same dimensions as  $\mathbf{E}$ . Compared to the equations in Zangwill, the overall sign differs here, and we have inserted a factor of  $k^{-1}$  to simplify the implementation. We can obtain  $c \mathbf{B}$  from  $\mathbf{E}$  by the substitution  $(u, w) \rightarrow (-w, u)$ ; we will use this relationship later to simplify the calculation. The scalar function  $u$  is given by

$$u = E_0 \sum_{\ell=1}^{\infty} i^{\ell} \frac{2\ell+1}{\ell(\ell+1)} f_{\ell}(k r) P_{\ell}^1(\cos \theta) \sin \phi. \quad (3)$$

Here,  $\theta$  and  $\phi$  are the usual spherical coordinates, and the  $P_{\ell}^1$  are associated Legendre polynomials. If  $f_{\ell}(k r) = j_{\ell}(k r)$ , where  $j_{\ell}(k r)$  is a spherical Bessel function, the expressions describe the incident fields for a plane wave in the case of an incident plane wave with spatial dependence  $\exp(i k z)$ . The sum of the incident and scattered fields is given by

$$f_{\ell}(k r) = j_{\ell}(k r) + a_{\ell} h_{\ell}^{(1)}(k r) \quad (4)$$

for  $u$ , where  $h_{\ell}^{(1)}(k r)$  is a spherical Hankel function. For  $w$ , we substitute  $a_{\ell} \rightarrow b_{\ell}$  and  $\sin \phi \rightarrow \cos \phi$  in the previous two expressions.

Explicit forms for the Mie coefficients  $a_{\ell}$  and  $b_{\ell}$  are given here for a particle with index of refraction  $n$  and radius  $a$ . We restrict attention to the case of a nonmagnetic dielectric sphere:

$$a_{\ell}(k a) = - \frac{j_{\ell}(n k a) S_{\ell}'(k a) - j_{\ell}(k a) S_{\ell}'(n k a)}{j_{\ell}(n k a) \xi_{\ell}(k a) - h_{\ell}^{(1)}(k a) S_{\ell}'(n k a)}, \quad (5)$$

$$b_{\ell}(k a) = - \frac{j_{\ell}(k a) S_{\ell}'(n k a) - n^2 j_{\ell}(n k a) S_{\ell}'(k a)}{h_{\ell}^{(1)}(n k a) S_{\ell}'(n k a) - n^2 j_{\ell}(n k a) S_{\ell}'(k a)}, \quad (6)$$

where  $S_\ell(x) = x j_\ell(x)$  and  $\xi_\ell(x) = x h_\ell^{(1)}(x)$  are Riccati–Bessel functions; a prime denotes differentiation with respect to the argument. The total scattering cross section is given by

$$\frac{\sigma_{\text{Mie}}}{\pi a^2} = \frac{2}{(ka)^2} \sum_{\ell=1}^{\infty} i^\ell (2\ell + 1) (|a_\ell|^2 + |b_\ell|^2), \quad (7)$$

written here in dimensionless form by incorporating the geometric cross section of the spherical particle; this form is commonly denoted by  $Q_{\text{sca}}$ .

## □ 1.1 Mie Coefficients and Cross Sections

The function `abCoefficients` calculates the Mie coefficients  $a_\ell$  and  $b_\ell$  using equations (5) and (6); the table of those coefficients is `abTable`. The functions `DxSphericalBesselJ` and `DxSphericalHankelH1` are  $S_\ell$  and  $\xi_\ell$ , respectively. The variable  $\ell$  (with maximum value `lMax`) is an index used for Legendre coefficients in physics. The variable `ka` is  $ka$ , the product of the wavevector  $k$  and the particle radius  $a$ ; `ka` ranges from `kaMin` to `kaMax` with step size `dka`. The function `MieCrossSection` returns a list pairing `ka` with the cross section normalized to the geometry. A simpler three-argument function calls a five-argument function. The parameter `convergenceFactor` helps to tell how many terms to calculate to achieve convergence; 1.5 seems to work well, but the reader may wish to test this.

```

DxSphericalBesselJ[l_, x_] :=
  Module[{y}, D[SphericalBesselJ[l, y], y] /. {y -> x}]

DxSphericalHankelH1[l_, x_] :=
  Module[{y}, D[SphericalHankelH1[l, y], y] /. {y -> x}]

abCoefficients[l_, nn_, ka_] := Module[
  {nka, aCoeff, bCoeff}
  , nka = nn ka
  ;
  aCoeff =
  - (SphericalBesselJ[l, nka] DxSphericalBesselJ[l, ka] -
    SphericalBesselJ[l, ka] DxSphericalBesselJ[l, nka]) /
  (SphericalBesselJ[l, nka] DxSphericalHankelH1[l, ka] -
    SphericalHankelH1[l, ka] DxSphericalBesselJ[l, nka])
  ;
  bCoeff =
  - (SphericalBesselJ[l, ka] DxSphericalBesselJ[l, nka] -
    nn^2 SphericalBesselJ[l, nka]
    DxSphericalBesselJ[l, ka]) /
  (SphericalHankelH1[l, ka] DxSphericalBesselJ[l, nka] -
    nn^2 SphericalBesselJ[l, nka]
    DxSphericalHankelH1[l, ka])
  ; {aCoeff, bCoeff}
]

```

```

MieCrossSection[kaMin_?NumericQ, kaMax_?NumericQ,
  dka_?NumericQ, nn_?NumericQ,
  convergenceFactor_?NumericQ] := Module[
  {fMax, abTable, ka}
  , Table[
    fMax = Ceiling[convergenceFactor nn kaMax]
    ; abTable = abCoefficients[Range[fMax], nn, ka]
    ;
    {ka,
      (2 / ka ^ 2)
      Total[(Re[abTable] ^ 2 + Im[abTable] ^ 2) .
        (2 Range[fMax] + 1)] (* dot sums over a and b *)
    }
  , {ka, kaMin, kaMax, dka}
  ]
]

MieCrossSection[kaMax_?NumericQ, dka_?NumericQ,
  nn_?NumericQ] :=
  Prepend[MieCrossSection[dka, kaMax, dka, nn, 1.5], {0, 0}]


```

Next, we plot the Mie cross section, similar to the one found in Zangwill [6] and Bohren and Huffman [5]. This is slow: it took 442 seconds on a 3.7 GHz personal computer. The code could be written to run significantly faster, but it would become more cryptic. The main point here is to verify the correctness of the code and to clarify the exposition. The red line is the asymptotic value for large  $ka$ , a dimensionless parameter comparing the particle size to the wavelength. The fact that this value is exactly twice the geometric cross section is discussed in [3, 4].

```

Module[
  {
    kaMax = 80. (* Use 80 to match Zangwill's plot *),
    dka = 0.2 (* step size *),
    nn = 1.33 (* index of refraction of water *)
  },
  ListLinePlot[
    MieCrossSection[kaMax, dka, nn]
    , FrameLabel -> {
      ka,
      Row[{Subscript["σ", "Mie"], " / π", a^2}]
    }
    , FrameStyle -> Thick
    , PlotRange -> {0, 4}
    , PlotStyle -> {Black}
    , Epilog -> {
      Inset[Row[{Style["n", Italic], " = ", nn}], {60, 3}],
      Red, Line[{{0, 2}, {kaMax, 2}}]
    }
  ]
]

```



▲ **Figure 1.** Mie cross section for a sphere of radius  $a$  and wavevector  $k$  compared to the geometric cross section and its large  $ka$  limit, in red. For the physical significance of the sharp peaks, see [7].

## □ 1.2 Expansion of Mie Coefficients for Small $ka$

Here we develop the Taylor expansion of the Mie coefficients for small  $ka$ . This subsection confirms equations (8–11). The results are used in Section 5.

We start with a simplification.

```
refine[expr_] := Refine[expr, {ka > 0, n > 0}]
```

The following definitions are motivated by equations (5) and (6).

```
aNumerator[l_, n_, ka_] :=
- (SphericalBesselJ[l, n ka] DxSphericalBesselJ[l, ka] -
  SphericalBesselJ[l, ka] DxSphericalBesselJ[l, n ka]) //
Simplify
```

```
aDenominator[l_, n_, ka_] :=
(SphericalBesselJ[l, n ka] DxSphericalHankelH1[l, ka] -
  SphericalHankelH1[l, ka] DxSphericalBesselJ[l, n ka]) //
Simplify
```

```
bNumerator[l_, n_, ka_] :=
- (SphericalBesselJ[l, ka] DxSphericalBesselJ[l, n ka] -
  n^2 SphericalBesselJ[l, n ka]
  DxSphericalBesselJ[l, ka]) // Simplify
```

```

bDenominator[ $\ell$ _, n_, ka_] :=
  (SphericalHankelH1[ $\ell$ , ka] DxSphericalBesselJ[ $\ell$ , n ka] -
   n^2 SphericalBesselJ[ $\ell$ , n ka]
   DxSphericalHankelH1[ $\ell$ , ka]) // Simplify

```

Later, we will find we need these for  $\ell = 1, 2$ . The index of refraction is  $n$ .

```

a1 = aNumerator[1, n, ka] / aDenominator[1, n, ka] //
  FullSimplify;

```

```

b1 = bNumerator[1, n, ka] / bDenominator[1, n, ka] //
  FullSimplify;

```

```

a2 = aNumerator[2, n, ka] / aDenominator[2, n, ka] //
  FullSimplify;

```

```

b2 = bNumerator[2, n, ka] / bDenominator[2, n, ka] //
  FullSimplify;

```

The Taylor series is taken next. A high order is necessary even for the small  $ka$  limit. In many cases, the spherical Neumann function enters the calculation, which leads to divergence in this limit. The real and imaginary parts are represented by the suffixes 1 or 2 in the variable names. The series expansions are chosen so as to include terms to the lowest nonvanishing order.

```

(*  $\ell=1$  *)
a1Series =
  Simplify[
    Normal[Series[a1, {ka, 0, 10},
      Assumptions  $\rightarrow$  {ka > 0, n > 0}]]];
b1Series =
  Simplify[
    Normal[Series[b1, {ka, 0, 6},
      Assumptions  $\rightarrow$  {ka > 0, n > 0}]]];

a1Re = Normal[Series[refine[Re[a1Series]], {ka, 0, 10},
  Assumptions  $\rightarrow$  {ka > 0, n > 0}]];
a1Im = Normal[Series[refine[Im[a1Series]], {ka, 0, 5},
  Assumptions  $\rightarrow$  {ka > 0, n > 0}]];
b1Re = Normal[Series[refine[Re[b1Series]], {ka, 0, 6},
  Assumptions  $\rightarrow$  {ka > 0, n > 0}]];
b1Im = Normal[Series[refine[Im[b1Series]], {ka, 0, 3},
  Assumptions  $\rightarrow$  {ka > 0, n > 0}]];

```

```

(* l=2 *)
a2Series =
  Simplify[
    Normal[Series[a2, {ka, 0, 14},
      Assumptions -> {ka > 0, n > 0}]]];
b2Series =
  Simplify[
    Normal[Series[b2, {ka, 0, 10},
      Assumptions -> {ka > 0, n > 0}]]];

a2Re = Normal[Series[refine[Re[a2Series]], {ka, 0, 14},
  Assumptions -> {ka > 0, n > 0}]];
a2Im = Normal[Series[refine[Im[a2Series]], {ka, 0, 7},
  Assumptions -> {ka > 0, n > 0}]];
b2Re = Normal[Series[refine[Re[b2Series]], {ka, 0, 10},
  Assumptions -> {ka > 0, n > 0}]];
b2Im = Normal[Series[refine[Im[b2Series]], {ka, 0, 5},
  Assumptions -> {ka > 0, n > 0}]];

TraditionalForm@TableForm@{
  {Row[{"a1Re = ", a1Re}], Row[{"a1Im = ", a1Im}]},
  {Row[{"b1Re = ", b1Re}], Row[{"b1Im = ", b1Im}]},
  {Row[{"a2Re = ", a2Re}], Row[{"a2Im = ", a2Im}]},
  {Row[{"b2Re = ", b2Re}], Row[{"b2Im = ", b2Im}]}
}

```

$$\begin{aligned}
 a1Re &= -\frac{ka^{10}(n^2-1)^2}{2025} & a1Im &= \frac{1}{45}ka^5(n^2-1) \\
 b1Re &= -\frac{4ka^6(n^2-1)^2}{9(n^2+2)^2} & b1Im &= \frac{2ka^3(n^2-1)}{3(n^2+2)} \\
 a2Re &= -\frac{ka^{14}(n^2-1)^2}{2480625} & a2Im &= \frac{ka^7(n^2-1)}{1575} \\
 b2Re &= -\frac{ka^{10}(n^2-1)^2}{225(2n^2+3)^2} & b2Im &= \frac{ka^5(n^2-1)}{15(2n^2+3)}
 \end{aligned}$$

The results are given here as mathematical formulas, which should agree with results in the cell above. Superscripts 1 and 2 refer to real and imaginary parts.

$$a_1^{(1)} = -\frac{(ka)^{10}}{2025}(n^2-1)^2 \quad a_1^{(2)} = \frac{1}{45}(ka)^5(n^2-1) \quad (8)$$

$$b_1^{(1)} = -\frac{4(ka)^6}{9}\left(\frac{n^2-1}{n^2+2}\right)^2 \quad b_1^{(2)} = \frac{2(ka)^3}{3}\left(\frac{n^2-1}{n^2+2}\right) \quad (9)$$

$$a_2^{(1)} = -\frac{(ka)^{14}}{2480625}(n^2-1)^2 \quad a_2^{(2)} = \frac{(ka)^7}{1575}(n^2-1) \quad (10)$$

$$b_2^{(1)} = -\frac{(ka)^{10}}{225}\left(\frac{n^2-1}{2n^2+3}\right)^2 \quad b_2^{(2)} = \frac{(ka)^5}{15}\left(\frac{n^2-1}{2n^2+3}\right) \quad (11)$$

## ■ 2. Maxwell Stress Tensor

For a harmonic electromagnetic field, the time-averaged electromagnetic force  $\langle \mathbf{F} \rangle$  on a dielectric particle in a vacuum is given by [1, 2, 4]

$$\langle \mathbf{F} \rangle = \oint dS \cdot \langle \mathbb{T} \rangle, \quad (12)$$

where the angle brackets mean the time average,  $dS$  is the differential of the surface normal, and the integral is taken over any surface enclosing the particle. Thus the Maxwell stress tensor is given by

$$\mathbb{T} = \mathbb{T}_E + \mathbb{T}_B, \quad (13)$$

where

$$\mathbb{T}_E = \epsilon_0 \left( \mathbf{E} \otimes \mathbf{E} - \frac{\mathbf{E} \cdot \mathbf{E}}{2} I \right), \quad (14)$$

and where  $\epsilon_0$  is the electric constant,  $\mathbf{E} \otimes \mathbf{E}$  is a dyadic, and  $I$  is the  $3 \times 3$  identity matrix. The companion matrix  $\mathbb{T}_B$  is given by  $\mathbb{T}_E$  with  $\mathbf{E} \rightarrow c \mathbf{B}$ . (The magnetic constant  $\mu_0$  used in the references is related by  $\mu_0^{-1} = \epsilon_0 c^2$ .)

Since the particle is spherical, it is natural to do the integral on the surface of a sphere at radius  $R \geq a$ . Since we are in a vacuum, the electromagnetic field exhibits no force there. Looking ahead, we will check to see that the result is independent of  $R$ . The direction of  $dS$  is  $\hat{\mathbf{r}}$ , so it is sufficient to calculate a mixed-basis tensor component  $\mathbb{T}_{ri}$ , where  $r$  refers to the component along  $\hat{\mathbf{r}}$  and  $i$  is a Cartesian direction. Rewriting equation (12) in spherical coordinates leads to

$$\langle F_i \rangle = R^2 \int_0^{2\pi} d\phi \int_0^\pi d\theta \sin \theta \langle \mathbb{T}_{r,i} \rangle. \quad (15)$$

We accept the complexity of the mixed-bases tensor  $\mathbb{T}$  because the dot product leads to a single component in spherical coordinates, but the integral requires coordinates that are independent of the point of integration. In practice, these are Cartesian. Looking ahead, we will see that only the  $z$  component will be nonzero, which further motivates the choice. The calculation will proceed by forming the electric field  $\mathbf{E}$  in spherical coordinates and making a row of  $\langle \mathbb{T}_{r,R} \rangle$ , where the index  $R$  implies spherical coordinates. The vector will be transformed to Cartesian coordinates by right-multiplying by a rotation matrix. (The same process is used for the magnetic field  $c \mathbf{B}$ .)

The effect of the time average is the following: given  $\mathbf{E}(\mathbf{r}, t) = \text{Re}[\mathbf{E}(\mathbf{r}) \exp(-i \omega t)]$ ,  $\langle \mathbf{E} \otimes \mathbf{E} \rangle = \frac{1}{4}[\mathbf{E}(\mathbf{r})^* \otimes \mathbf{E}(\mathbf{r}) + \mathbf{E}(\mathbf{r}) \otimes \mathbf{E}(\mathbf{r})^*]$ . The stress tensor will be computed from the second term. The first term is included later by adding the complex conjugate.



### ■ 3. Scattering Force

For the scattering force, we take the incident electric field to be a plane wave traveling in the positive  $z$  direction and linearly polarized in the  $x$  direction. Explicitly,

$$\mathbf{E}(\mathbf{r}, t) = E_0 \hat{\mathbf{x}} \exp(ikz) \cos(\omega t). \quad (16)$$

#### □ Code: Symbolic Algebra

As discussed, we need to transform a vector from spherical to Cartesian coordinates. We give that function first.

```

CartesianFromSpherical[vec_] :=
vec.Transpose@{
  {Sin[ $\theta$ ] Cos[ $\phi$ ], Cos[ $\theta$ ] Cos[ $\phi$ ], -Sin[ $\phi$ ]}
  , {Sin[ $\theta$ ] Sin[ $\phi$ ], Cos[ $\theta$ ] Sin[ $\phi$ ], Cos[ $\phi$ ]}
  , {Cos[ $\theta$ ], -Sin[ $\theta$ ], 0}
}

```

The electric or magnetic stress tensor is computed from the following code. `timeAverage` is a factor required for the time average.

```

timeAverage = 1 / 4;
StressTensorRow[field_] :=
timeAverage
  (First[field] conj[field] +
   (field . conj[field]) {-1 / 2, 0, 0})

```

`First[field]` gets the  $r$  component, since the field components have the usual order ( $r, \theta, \phi$ ); `conj[field]` is the complex conjugate, but we need to help the kernel find simplifications. (The magnetic part will be found by symbolic substitution.) The conjugation is performed as follows.

```

conj[vec_] := Conjugate[vec] // conjugateExpand // reExpand //
  Refine[#, {0 ≤  $\theta$  ≤ Pi, Element[ $\phi$ , Reals],
    Element[kr, Reals]}] &

conjugateExpand[expr_] := Module[
  {exprB, exprC}
  , expr //. {
    Conjugate[exprB + exprC] :>
      Conjugate[exprB] + Conjugate[exprC]
    , Conjugate[exprB * exprC] :>
      Conjugate[exprB] * Conjugate[exprC]
  }
]

```

reExpand is used for real expansion.

```
reExpand[expr_] :=
Module[
  {exprB, exprC}
  , expr //. {
    Re[exprB_ + exprC_] := Re[exprB] + Re[exprC]
    , Re[exprB_ * exprC_] := Re[exprB] * Re[exprC]
  }
]
```

Following are some additional simplifications that we will use. We operate in the dimensionless radial variable called  $kr$ , which is  $kr$  (i.e.  $k$  times  $r$ ), where  $k$  and  $r$  do not enter the calculation independently. The assumptions are made because of the range of integration.

```
assumptions = {0 ≤ θ ≤ Pi, Im[φ] == 0, kr > 0};
simplify[expr_] := Simplify[expr, Assumptions → assumptions]
```

We introduced extra factors of  $k$  into equations (1) and (2) because we implement  $k^{-1} \nabla \times$  rather than  $\nabla \times$  to enable the use of  $kr$ . The operation  $k^{-1} \nabla \times$  is named `kurl`, the curl for a dimensionless radial variable.

```
kurl[vec_] := Curl[vec, {kr, θ, φ}, "Spherical"]
```

Next we define a function to create the electric and magnetic fields (“em”), omitting the prefactor  $E_0$ . Before the final answer for the force is obtained, the factor will be reintroduced as the variable  $E0$ .

```
emMie[ℓMax_, zDir_, useExt_, useInd_] := Module[
  {uTot, wTot, flipYZ, rVecU, rVecW, curlRVecU,
   curlRVecW, curlCurlRVecU, curlCurlRVecW}
  , uTot = Total@Table[
    simplify[I^ℓ ((2ℓ + 1) / ((ℓ (ℓ + 1))))
      (useExt j[ℓ, kr] + useInd a[ℓ] h1[ℓ, kr])
      LegendreP[ℓ, 1, Cos[θ]] Sin[φ]]
    , {ℓ, ℓMax}
  ]
  ; wTot = uTot /. {a → b, Sin[φ] → Cos[φ]}
  ; If[zDir == -1
    (* backward-going wave *)
    , flipYZ = {φ → -φ, θ → Pi - θ}
    (* 180 degree rotation about x in spherical
      coordinates *)
    ; uTot = uTot /. flipYZ
    ; wTot = wTot /. flipYZ
  ]
  ;
  rVecU = {kr uTot, 0, 0}
  (* vector in spherical coordinates *)
  ; rVecW = {kr wTot, 0, 0}
  ; curlRVecU = simplify[kurl[rVecU]]
```

```

; curlRVecW = simplify[kurl[rVecW]]
; curlCurlRVecU = simplify[kurl[curlRVecU]]
; curlCurlRVecW = simplify[kurl[curlRVecW]]
;
{
  simplify[+curlRVecU - I curlCurlRVecW],
  simplify[-curlRVecW - I curlCurlRVecU]
}
]

```

This gives a simplified interface for a positive  $z$  plane wave.

```
emMiePositiveZ[?Max_?IntegerQ] := emMie[?Max, 1, 1, 1]
```

It is sufficient to set `?Max = 2` for both the scattering and gradient forces, since both occur in the limit of small particles. Setting `?Max = 1` does not capture all contributing terms to lowest order, and setting `?Max = 3` or higher does not cause the limit to change. For the scattering, we will have `zDir = 1`, which is for a plane wave going in the positive  $z$  direction. For the gradient force where we have a standing wave, we decompose that into the sum of a plane wave going toward  $+z$  and one going toward  $-z$ . The backward wave is still polarized in the  $x$  direction, so we produce it by mapping  $(x, y, z) \rightarrow (x, -y, -z)$ . Flipping  $y$  is necessary because the direction of travel is  $\mathbf{E} \times \mathbf{B}$ ; if we wish to preserve  $\mathbf{E}$ , it is necessary to change the sign of  $\mathbf{B}$ . We will always set `useExt = 1` and `useInd = 1`; these may be set to zero to suppress the external field or the induced field, respectively.

We next determine the electric and magnetic fields. Although we could increase `?Max`, the run time increases dramatically. For example, the case of `?Max = 6` from [1, 2] was run overnight.

For `?Max`, use 1 for a quick test, 2 to get the main results, and 3 or more to confirm that a higher-order expansion does not change the limit. The run time increases rapidly as `?Max` is increased. Run times can vary, but typically it is best to start the code and come back to it after a few minutes to a couple of hours. The parameter `?Max` is used both here and for the gradient force in Section 4.

```
?Max = 2;
```

Next, we make rows from the electric part and magnetic part (multiplied by  $c$ ) of the stress tensor in Cartesian coordinates.

```

{electricStressTensorRow, cMagneticStressTensorRow} =
  simplify[CartesianFromSpherical[StressTensorRow[#]]] & /@
  emMiePositiveZ[?Max];

```

We do the integrals over  $\phi$  next, yielding the three Cartesian components of the electric part of the stress tensor, after azimuthal integration.

The azimuthal integrals of the  $x$  and  $y$  components of the electric part of the stress tensor are zero, as they should be. The  $z$  component is nonzero in general and will be used later.

```
Integrate[Expand[electricStressTensorRow[[1]],
  { $\phi$ , 0, 2 Pi}]
```

```
0
```

```
Integrate[Expand[electricStressTensorRow[[2]],
  { $\phi$ , 0, 2 Pi}]
```

```
0
```

```
 $\phi$ ElectricIntegralZ =
  Integrate[Expand[electricStressTensorRow[[3]],
    { $\phi$ , 0, 2 Pi}] // simplify;
```

The magnetic terms are similar.

```
Integrate[Expand[cMagneticStressTensorRow[[1]],
  { $\phi$ , 0, 2 Pi}]
```

```
0
```

```
Integrate[Expand[cMagneticStressTensorRow[[2]],
  { $\phi$ , 0, 2 Pi}]
```

```
0
```

```
 $\phi$ cMagneticIntegralZ =
  Integrate[Expand[cMagneticStressTensorRow[[3]],
    { $\phi$ , 0, 2 Pi}] // simplify;
```

The integral over  $\theta$  is done next. Although these variables contain the word `Force`, they lack some constant factors to be forces, hence the prefix `pre`. These factors will be included after a few manipulations. The notation ending with `Z` is for the  $z$  component. The successive variables `preForceZ` lack the constant, but the final result `ForceZ` is the force.

```
preElectricForceZ[1] =
  Integrate[Expand[Sin[ $\theta$ ]  $\phi$ ElectricIntegralZ], { $\theta$ , 0, Pi},
  Assumptions  $\rightarrow$  assumptions] // simplify;
```

```
preMagneticForceZ[1] =
  Integrate[Expand[Sin[ $\theta$ ]  $\phi$ cMagneticIntegralZ],
  { $\theta$ , 0, Pi}, Assumptions  $\rightarrow$  assumptions] // simplify;
```

Combining the electric and magnetic terms into a total force at this point leads to some cancellations, so it is convenient to do it next.

```
preForceZ[1] = preElectricForceZ[1] + precMagneticForceZ[1] //  
simplify;
```

Up to this point, we have done the manipulations without telling the kernel that  $j$  and  $h_1$  are, in fact, spherical Bessel functions  $j_\ell$  and  $h_\ell^{(1)}$ . Moreover, we use  $h_\ell^{(1)} = j_\ell + i y_\ell$  (a complex function of real arguments) because we will be obtaining real and imaginary parts shortly.

```
j[l_, kr_] := SphericalBesselJ[l, kr]
```

```
h1[l_, kr_] := SphericalBesselJ[l, kr] +  
I SphericalBesselY[l, kr]
```

```
preForceZ[2] = preForceZ[1] // conjugateExpand // simplify;
```

As discussed previously, we have calculated only one of two terms in the force. Next, we add the complex conjugate.

```
preForceZ[3] =  
(preForceZ[2] + Conjugate[preForceZ[2]]) //  
conjugateExpand // simplify;
```

We introduce real and imaginary parts for the Mie coefficients.

```
preForceZ[4] = preForceZ[3] /.  
{  
  a[l_] -> aRe[l] + I aIm[l]  
  , b[l_] -> bRe[l] + I bIm[l]  
} // conjugateExpand // Refine[#, {  
  Element[aRe[l_], Reals]  
  , Element[aIm[l_], Reals]  
  , Element[bRe[l_], Reals]  
  , Element[bIm[l_], Reals]}} & // simplify;
```

The expression is still quite complicated. However, `FullSimplify` knows (for  $\ell \leq 9$ ) that the Wronskian of the spherical Bessel functions  $W(j_\ell, y_\ell) = (kr)^{-2}$  [8]. We will see that the force does not depend on the radius  $r$  of the integration sphere, as long as  $r \geq a$ . We expect this on physical grounds: there should be no force on the vacuum region surrounding the physical sphere. Mathematically, the Wronskian plays a key role in achieving this condition.

```
preForceZ[5] = preForceZ[4] // FullSimplify;
```

We next include some constant prefactors, namely  $\epsilon_0$ ,  $E_0^2$  and  $r^2$ . The factors  $\epsilon_0$  and  $E_0^2$  arise from the definition of the stress tensor. The factor  $r^2$  comes from the fact that when we performed the two angular integrals over the sphere, we did not yet include the dimensional constant associated with the area element. This extra factor of  $r^2$  is the only exception to the radius appearing in the dimensionless variable  $kr$ . The substitution allows powers of  $kr$  to be canceled.

$$\text{forceZ} = \epsilon_0 E_0^2 r^2 \text{preForceZ}[5] /. \{\mathbf{r} \rightarrow \mathbf{kr} / \mathbf{k}\}$$

$$-\frac{1}{3 k^2} E_0^2 \pi \epsilon_0 (7 \text{aRe}[2] + 9 \text{aIm}[1] (\text{aIm}[2] + \text{bIm}[1]) + 5 \text{aIm}[2] \text{bIm}[2] + 9 \text{bIm}[1] \text{bIm}[2] + 9 \text{bRe}[1] + 9 \text{aRe}[1] (1 + \text{aRe}[2] + \text{bRe}[1]) + (7 + 5 \text{aRe}[2] + 9 \text{bRe}[1]) \text{bRe}[2])$$

The real and imaginary parts of the Mie coefficient  $a_\ell$  are  $\text{aRe}[\ell]$  and  $\text{aIm}[\ell]$ , respectively, and similarly, for  $b_\ell$  they are  $\text{bRe}[\ell]$  and  $\text{bIm}[\ell]$ .

## ■ 4. Gradient Force

In the previous section, we used the general form for the electric and magnetic fields for the Mie expansion as input to the Maxwell stress tensor to derive the force on a particle. We do the same in this section, borrowing from the previous section to the extent possible. However, the spherical particle is in a standing wave field instead of a plane wave traveling in the  $+z$  direction. The standing wave is a linear superposition of two plane waves going toward  $+z$  and toward  $-z$ , so we need a solution from the latter field. We obtain this by symmetry from the existing solution. Sending  $(x, y, z) \rightarrow (x, -y, -z)$  does what is needed: the polarization of  $\mathbf{E}$  is unchanged, but the direction of propagation  $\hat{\mathbf{E}} \times \hat{\mathbf{B}}$  changes sign. This is implemented with the wrapper to the function `emMie` described in Section 3.

$$\text{emMieNegativeZ}[\text{?Max\_?IntegerQ}] := \text{emMie}[\text{?Max}, -1, 1, 1]$$

Although the standing wave is the sum of two plane waves, more precisely, a phased sum is required. Although we do not show it here, the answer is proportional to  $\text{Sin}[2 \text{phase}]$ . This is because the dielectric sphere is located at the coordinate origin, and we need to have a maximum of the electric field there. We do this as follows. (In the variable names,  $\nabla$  stands for gradient.)

$$\text{phase} = \text{Pi} / 4;$$

$$\{\text{Electric}\nabla, \text{cMagnetic}\nabla\} =$$

$$\text{Apart}[\text{emMiePositiveZ}[\text{?Max}] \text{Exp}[\text{I phase}]] +$$

$$\text{Apart}[\text{emMieNegativeZ}[\text{?Max}] \text{Exp}[-\text{I phase}]];$$

From here, the manipulation is the same as for the scattering force, so the code is written without further explanation. Certain terms integrate to zero.

First, the electric terms in the stress tensor are calculated. As previously, the  $x$  and  $y$  components are 0.

```

electricStressTensor $\nabla$  =
  CartesianFromSpherical [StressTensorRow [Electric $\nabla$ ]] //
  simplify;

cMagneticStressTensor $\nabla$  =
  CartesianFromSpherical [StressTensorRow [cMagnetic $\nabla$ ]] //
  simplify;

Integrate [Expand [electricStressTensor $\nabla$ [[1]]], { $\phi$ , 0, 2 Pi}]
0

Integrate [Expand [electricStressTensor $\nabla$ [[2]]], { $\phi$ , 0, 2 Pi}]
0

 $\phi$ ElectricIntegral $\nabla$  =
  Integrate [Expand [electricStressTensor $\nabla$ [[3]]], { $\phi$ , 0, 2 Pi}] //
  simplify;

precMagneticForce $\nabla$ [1] =
  Integrate [Expand [Sin [ $\theta$ ]  $\phi$ ElectricIntegral $\nabla$ ],
    { $\theta$ , 0, Pi}, Assumptions  $\rightarrow$  assumptions] // simplify;

```

Second, the magnetic terms in the stress tensor are calculated. Again, the  $x$  and  $y$  components are 0 and the  $z$  component is nonzero in general.

```

Integrate [Expand [cMagneticStressTensor $\nabla$ [[1]]], { $\phi$ , 0, 2 Pi}]
0

Integrate [Expand [cMagneticStressTensor $\nabla$ [[2]]], { $\phi$ , 0, 2 Pi}]
0

 $\phi$ cMagneticIntegral $\nabla$  =
  Integrate [Expand [cMagneticStressTensor $\nabla$ [[3]]],
    { $\phi$ , 0, 2 Pi}] // simplify;

cMagneticStressTensor $\nabla$  =
  Integrate [Expand [Sin [ $\theta$ ]  $\phi$ cMagneticIntegral $\nabla$ ],
    { $\theta$ , 0, Pi}, Assumptions  $\rightarrow$  assumptions] // simplify;

```

Finally, the electric and magnetic terms are combined and the force is found.

```

preForceZ∇[1] =
  precMagneticForceZ∇[1] + cMagneticStressTensor∇ // simplify;

preForceZ∇[2] = preForceZ∇[1] // conjugateExpand // simplify;

preForceZ∇[3] =
  (preForceZ∇[2] + Conjugate[preForceZ∇[2]]) //
  conjugateExpand // simplify;

preForceZ∇[4] = preForceZ∇[3] /.
  {a[l_] → aRe[l] + I aIm[l]
   , b[l_] → bRe[l] + I bIm[l]
  } // conjugateExpand // Refine[#, {
    Element[aRe[l_], Reals]
    , Element[aIm[l_], Reals]
    , Element[bRe[l_], Reals]
    , Element[bIm[l_], Reals]}] & // simplify;

preForceZ∇[5] = preForceZ∇[4] // FullSimplify;

forceZ∇ = ε0 E0^2 r^2 preForceZ∇[5] /. {r → kr / k}

1
3 k^2 2 E0^2 π ε0
(9 aIm[1] (1 + aRe[2] + bRe[1]) + bIm[2] (7 + 5 aRe[2] + 9 bRe[1]) -
  9 bIm[1] (1 + aRe[1] + bRe[2]) -
  aIm[2] (7 + 9 aRe[1] + 5 bRe[2]))

```



## ■ 5. Forces in the Limit of Small $ka$

Our next task is to determine the scattering and gradient forces on spheres up to the leading order in  $ka \ll 1$ . Physically, these are spheres that are small compared to a wavelength. We combine the results of Section 3 for the scattering force and Section 4 for the gradient force with those of Section 2 for the Mie coefficients. In Section 1.2, we showed that  $a_1^{(1)} \sim (kr)^{10}$ ,  $a_1^{(2)} \sim (kr)^5$ ,  $b_1^{(1)} \sim (kr)^6$ ,  $b_1^{(2)} \sim (kr)^3$ ,  $a_2^{(1)} \sim (kr)^{14}$ ,  $a_2^{(2)} \sim (kr)^7$ ,  $b_2^{(1)} \sim (kr)^{10}$  and  $b_2^{(2)} \sim (kr)^5$  using superscripts (1) and (2) for the real and imaginary parts, respectively. Therefore, the scattering force, to lowest order in  $kr$ , retains only the terms in  $b_1^{(1)}$ , and for the gradient force, the lowest-order coefficient is  $b_1^{(2)}$ . The terms were given in equation (9). The key point is that the Clausius–Mossotti factor  $\frac{n^2-1}{n^2+2}$  appears with the second and first powers in the two terms. The result falls out of a Taylor expansion without any appeal to physical arguments about the response of dipoles. All of the terms are proportional to  $n^2 - 1$ , leading to the physically required result that if the sphere in fact contains a vacuum ( $n = 1$ ), there is no electromagnetic response and thus no force. However, the denominator  $n^2 + 2$  is characteristic only of the terms giving the lowest-order response. Higher terms have different dependencies on the index of refraction  $n$ , such as  $\frac{n^2-1}{2n^2+3}$ , as seen previously. The gradient and scattering forces properly exist only in the limit of small  $ka$ .

Having selected the lowest-order terms for the two force expansions, these are the forces to lowest order in  $ka$ .

$$F^{(\text{scat})} = -\frac{3\pi\epsilon_0 E_0^2}{k^2} b_1^{(1)} = \frac{4\pi\epsilon_0 E_0^2}{3k^2} \left(\frac{n^2-1}{n^2+2}\right)^2 (ka)^6, \quad (17)$$

$$F^{(\text{grad})} = -\frac{6\pi\epsilon_0 E_0^2}{k^2} b_1^{(2)} = \frac{4\pi\epsilon_0 E_0^2}{k^2} \left(\frac{n^2-1}{n^2+2}\right) (ka)^3. \quad (18)$$

These formulas agree with those given by Harada and Asakura [3], as shown in the Appendix.

## ■ Summary

Our goal was to derive the scattering force and the gradient force rigorously from the Mie solution and the Maxwell stress tensor. We began by presenting the Mie solution for a plane wave incident on a dielectric sphere in a vacuum and showing that our implementation matches a figure from a textbook. We then presented the formula for the force in terms of a surface integral of the Maxwell stress tensor, which we take on a sphere of arbitrary radius centered on and including the whole dielectric sphere. We analyzed the scattering force first, giving a formula for the force in terms of the Mie coefficients and then taking the limit as the radius of the sphere tends to zero. This yields agreement with the usual formula for the scattering force in a vacuum. The forces were reformulated for a standing wave, and a similar program was carried out, leading to agreement for the gradient force.

## ■ Conclusion

The Clausius–Mossotti term that appears in expressions for the scattering force and the gradient force is seen to be implicit in the rigorous Mie solutions to Maxwell’s equations. By finding the Maxwell stress tensor for a plane wave or a standing wave acting on a dielectric sphere, we are able to show the response in lowest order is in agreement with a widely used formula for the scattering force and the gradient force, respectively. Since part of the derivation includes a tenth-order Taylor expansion of special functions, it is difficult to see how the result could be obtained without computer-assisted algebra.

## ■ Appendix: Reconciling the Harada–Asakura Formulas with the Present Results

First, we wish to match equation (17) to the scattering force as given in equation (12) of [3], hereafter called equation (HA12). Variables with superscript (HA) are from the reference. We consider only the case of the external medium being a vacuum, so  $n_2^{(\text{HA})} = 1$ . This implies  $m^{(\text{HA})} = n_1^{(\text{HA})} = n$ , the index in the sphere, so that the Clausius–Mossotti factor is present in both our equation (17) and equation (HA12). The same holds for equation (18) and equation (HA16). Next, [3] considers a Gaussian beam profile, so we simply pick the point in the middle, setting  $\tilde{x}^{(\text{HA})} = \tilde{y}^{(\text{HA})} = \tilde{z}^{(\text{HA})} = 0$ . The factor  $\frac{2P^{(\text{HA})}}{\pi(w_0^{(\text{HA})})^2}$  is the intensity at the center of the beam  $I$ . The beam intensity is related to the electrical field by  $\frac{I}{c} = \frac{1}{2} \epsilon_0 E_0^2$ . Given these expressions, all the factors can be matched by inspection.

Next, we wish to match equation (18) to the gradient force given in equation (HA16). The one additional equation to note is that  $\frac{1}{c} \nabla I^{(\text{HA})}(r) = \hat{z} \frac{2}{c} \epsilon_0 E_0^2 2k \sin(2kz) |_{kz=\pi/4}$ . The factor of 2 occurs in the numerator because, by our definition,  $E_0$  represents the field of one of the two interfering beams. The 2 in the denominator appears in the intensity-field conversion equation. The twos in  $2k \sin(2kz)$  are due to the physical fact that a standing plane wave interference pattern is periodic with half the wavelength of the electric fields of the plane waves of which the interference pattern is composed. Again, a match can be made by inspection.

## ■ Acknowledgments

Eric L. Shirley provided a key step in the derivation.

## ■ References

- [1] J. J. Curry and Z. H. Levine, "Continuous-Feed Optical Sorting of Aerosol Particles," *Optics Express*, **24**(13), 2016 pp. 14100–14123. doi:10.1364/OE.24.014100.
- [2] Z. H. Levine and J. J. Curry, "OptSortSph: Optical Sorting in a Standing Wave Field Calculated with Effective Velocities and Diffusion Constants," *Journal of Research of the National Institute of Standards and Technology*, **121**, 2016 pp. 420–421. doi:10.6028/jres.121.020.
- [3] Y. Harada and T. Asakura, "Radiation Forces on a Dielectric Sphere in the Rayleigh Scattering Regime," *Optics Communications*, **124**(5-6), 1996 pp. 529–541. doi:10.1016/0030-4018(95)00753-9.
- [4] G. Mie, "Beiträge zur Optik trüber Medien, speziell kolloidaler Metallösungen," *Annalen der Physik*, **330**(3), 1908 pp. 377–445. [onlinelibrary.wiley.com/doi/10.1002/andp.19083300302/pdf](http://onlinelibrary.wiley.com/doi/10.1002/andp.19083300302/pdf).
- [5] C. F. Bohren and D. R. Huffman, *Absorption and Scattering of Light by Small Particles*, New York: Wiley, 1998.
- [6] A. Zangwill, *Modern Electrodynamics*, Cambridge: Cambridge University Press, 2013.
- [7] P. Chylek, J. T. Kiehl and M. K. W. Ko, "Optical Levitation and Partial-Wave Resonances," *Physical Review A*, **18**(5), 1978 pp. 2229–2233. doi:10.1103/PhysRevA.18.2229.
- [8] M. Abramowitz and I. A. Stegun, *Handbook of Mathematical Functions*, 10th printing, New York: Dover, 1972, Equation (10.1.6).

Z. H. Levine and J. J. Curry, "Scattering and Gradient Forces from the Electromagnetic Stress Tensor Acting on a Dielectric Sphere," *The Mathematica Journal*, 2017. dx.doi.org/doi:10.3888/tmj.19-1.

## About the Authors

Zachary H. Levine is a physicist at NIST and a Fellow of the American Physical Society. He has concentrated on various aspects of the interaction of light and matter, including the calculation of dielectric constants, second harmonic coefficients and higher optical tensors.

J. J. Curry is an electrical engineer at the National Institute of Standards and Technology. His current interests include manipulation and measurement of particles using optical forces. In recent years, he has worked on problems of interest to the lighting industry.

*Zachary H. Levine*  
*The National Institute of Standards and Technology*  
*Gaithersburg, MD 20899-8410*  
*zlevine@nist.gov*

*J. J. Curry*  
*The National Institute of Standards and Technology*  
*Gaithersburg, MD 20899-8441*  
*john.curry@nist.gov*